APPROVED BY
J N Snyder

IDENTIFICATION

Read Octal Number Routine, MU RON1

M. R. Storm - January 17, 1957

Midwestern Universities Research Association, Madison, Wisconsin

PURPOSE

To read octal locations and numbers from 704 cards, convert to binary, and place the converted words into their specified locations.

RESTRICTIONS

Any number of octal words up to four are allowed per card. The specified location immediately precedes each 12-character word. When placing octal data into consecutive memory locations only the first needs to be punched on the card. The program loads cards sequentially until it encounters another location and proceeds from there. Blank 12-character words are ignored.

The information is punched into 68 columns of 1-72 in the following manner:

(a) Column 1 must be blank.
(b) Columns 2-6 may be any 5-character octal address greater than $(00142)_8$, when used as a self-loading program.
(c) Columns 7-18 are the first 12-character octal number.
(d) Column 19 must be left blank.
(e) Columns 20-24 are the second 5-character octal address.
(f) Columns 25-36 are the second 12-character octal number.

The right half of the card is punched precisely in the same order as the left half. Minus signs may be over punched in columns 7, 25, 43, or 61.

METHOD

Accuracy: Exact

USAGE

The normal usage of this program was designed as an auxiliary routine for loading octal numbers and corrections. In that sense it is a self-loading program and can be used alone or in conjunction with other loading programs. It recognizes a transfer card or another self-loading program and relinquishes control to it. As a self-loading program this routine occupies locations $0\text{-}98_{10}$ $(0\text{-}142_8)$. In use the program is backed by the octal cards to be loaded and the LOAD CARDS button pressed.

The additional feature of using this program as a closed input subroutine is provided but not encouraged. One accomplishes this technique by removing cards 1 through 12 of the symbolic deck and the END card (#94) (i.e. the self-loading portion), and prefixing the BSS card provided (#95). (This BSS card should be completely removed when assembling the self-loading program from the symbolic cards using the UA SAP twenty-four word per card option). The calling sequence then becomes:

UA SAP		MURASS	
Loc.	Instruction	Loc.	Instruction
α	TSX RON1, 4	α	B TSX A 4

Exit from the routine when used either as a self-loading routine or as a closed subroutine is made as follows:

Upon encountering any punching in 9 left of a data card the routine will transfer control to location 0, which already contains the 9 left information. Hence any valid 704 instruction punched into this location will be executed. If the transfer card or any card with 9 left punched is omitted, an end-of-file condition will produce a read-write check at RON1 + 10. (34_8 when used as a self-loading program.)

Space Required:

99 words including temporary storage.

	UA SAP	MURASS
81 words program at	RON1	A
2 words of fixed temporary at	0	0
16 words of temporary are also fixed at	RON1 - 16	A - 16

(Note: In a MURASS assembly only the symbol A needs to be defined; the location of temporary storage is fixed at 0 and A - 16 as indicated above.

In a UA SAP assembly the block of 16 storage locations will be automatically provided just before RON1.)

Error Codes:

The program checks for illegal multiply-punched columns and for non-octal characters exclusive of 9 left punches. The finding of such characters causes a stop (HTR RON1 + 2) at RON1 + 49.

CODING INFORMATION

Timing: The card reader rate of 250 cards per minute is maintained.

```
      REM  RON1 MURA READ OCTAL NUMBER                          MU   RON1
      LXA  0,4              CLEAR X4                            RON1 0001
      CPY  2,4              COPY FIRST CARD                     RON1 0002
      TXI  1,4,-1           DECREASE X4 BY 1                    RON1 0003
      HTR  75                                                   RON1 0004
      LTM                   TURN OFF TRAP INDICATOR             RON1 0005
      LXA  3,4              7 TO X4                             RON1 0006
      RCD                   SELECT CARD READER                  RON1 0007
      CPY  99,4             COPY REMAINING CARDS                RON1 0008
      TIX  7,4,1            OF PROGRAM                          RON1 0009
      TRA  RON1             PROGRAM LOADED                      RON1 0010
      TRA  6                END OF RECORD                       RON1 0011
      OCT  0,,,,,,,                                             RON1 0012
RON1  SXD  RON1+77,1        SAVE CONTENTS X1                    RON1 0013
      SXD  RON1+78,2        SAVE CONTENTS X2                    RON1 0014
      RCD                   SELECT CARD READER                  RON1 0015
      LXA  RON1+77,4        I=16 TO X4                          RON1 0016
      HTR  RON1,4,24576     0 TO (RON1-I)                       RON1 0017
      TIX  RON1+4,4,1       COMPARE AND DECREASE I              RON1 0018
      LXA  RON1+78,1        SET CARD ROW COUNT J                RON1 0019
      LXA  RON1+77,2        SET OCTAL WORD COUNT K              RON1 0020
      CPY  0                LEFT ROW TO 0                       RON1 0021
      TXH  RON1+65,1,10     CHECK FOR CHARACTER NOT OCTAL       RON1 0022
      CPY  1                RIGHT ROW TO 1                      RON1 0023
      TXH  RON1+73,1,10                                         RON1 0024
      LXA  RON1+13,4        SET LOCATION COLUMN COUNT =5        RON1 0025
      PXD  5                CLEAR AC                            RON1 0026
      LDQ  0                JTH ROW TO MQ                       RON1 0027
      ALS  2                CONVERT JTH LOCATION ROW            RON1 0028
      LLS  1                X                                   RON1 0029
      TIX  RON1+15,4,1      DECREASE LOCATION COLUMN COUNTER    RON1 0030
      TXL  RON1+24,1,2      TEST FOR 11 ROW                     RON1 0031
      ACL  RON1,2           FORM COMPARAND                      RON1 0032
      SLW  RON1,2           FOR DOUBLE-PUNCHED COLUMN           RON1 0033
      TXL  RON1+24,1,3      TEST FOR ZERO ROW                   RON1 0034
      ACL  RON1+1,2         ACCUMULATE CONVERTED ROWS           RON1 0035
      SLW  RON1+1,2         X                                   RON1 0036
      LXA  RON1+78,4        SET WORD COLUMN COUNT =12           RON1 0037
      ALS  2                CONVERT JTH ROW OF WORD             RON1 0038
      LLS  1                X                                   RON1 0039
      TIX  RON1+25,4,1      DECREASE WORD COLUMN COUNTER        RON1 0040
      TXL  RON1+34,1,2      TEST FOR 11 ROW                     RON1 0041
      ACL  RON1+2,2         FORM COMPARAND                      RON1 0042
      SLW  RON1+2,2         X                                   RON1 0043
      TXL  RON1+34,1,3      TEST FOR ZERO ROW                   RON1 0044
      ACL  RON1+3,2         ACCUMULATE CONVERTED ROWS           RON1 0045
      SLW  RON1+3,2         X                                   RON1 0046
      TXH  RON1+37,1,2      TEST FOR 11 ROW                     RON1 0047
      ALS  2                OVERFLOW                            RON1 0048
      ORS  RON1+3,2         PREFIX SIGN                         RON1 0049
      CLA  0                PREPARE TO CONVERT JTH ROW          RON1 0050
      LDQ  1                OF NEXT LOCATION AND WORD           RON1 0051
      LGL  18               X                                   RON1 0052
      STO  0                X                                   RON1 0053
```

```
STQ 1                   X                                   RON1  0054
TIX RON1+12,2,4         DECREASE WORD COUNT K               RON1  0055
TIX RON1+7,1,1          REDUCE ROW COUNT J                  RON1  0056
LXA RON1+77,4           RESTORE K=16 TO X4                  RON1  0057
CLA RON1,4              DOES LOCATION = ZERO                RON1  0058
TZE RON1+50             X                                   RON1  0059
SUB RON1+79             DOUBLE-PUNCH CHECK                  RON1  0060
TZE RON1+50             X                                   RON1  0061
HTR RON1+2              ERROR STOP                          RON1  0062
CLA RON1+2,4            DOES WORD = ZERO                    RON1  0063
TZE RON1+63             X                                   RON1  0064
SUB RON1+80             DOUBLE-PUNCH CHECK                  RON1  0065
TNZ RON1+49             X                                   RON1  0066
CLA RON1+1,4            IS LOCATION ZERO                    RON1  0067
TZE RON1+58             X                                   RON1  0068
SUB RON1+76             PLANT LOCATION                      RON1  0069
STA RON1+62             X                                   RON1  0070
CLA RON1+62             X                                   RON1  0071
ADD RON1+76             X                                   RON1  0072
STA RON1+62             X                                   RON1  0073
CLA RON1+3,4                                                RON1  0074
STO                     STORE CONVERTED WORD                RON1  0075
TIX RON1+45,4,4         DECREASE K                          RON1  0076
TRA RON1+2              READ NEXT CARD                      RON1  0077
LLS 37                  CONTENTS OF MQ TO AC                RON1  0078
TZE RON1+10             TEST FOR NON-OCTAL CHARACTER        RON1  0079
TXL RON1+49,1,11        X                                   RON1  0080
TOV RON1+69             OVERFLOW OFF                        RON1  0081
LXD RON1+77,1           RESTORE X1                          RON1  0082
LXD RON1+78,2           RESTORE X2                          RON1  0083
CPY 1                   X                                   RON1  0084
TRA 0                   EXIT TO 0                           RON1  0085
LLS 37                  TEST RIGHT COPY                     RON1  0086
TZE RON1+12             FOR NON-OCTAL CHARACTER             RON1  0087
TRA RON1+49             X                                   RON1  0088
HTR 1                   1B35                                RON1  0089
HTR 16                  16B35                               RON1  0090
HTR 12                  12B35                               RON1  0091
OCT 000000011111                                            RON1  0092
OCT 111111111111                                            RON1  0093
END                                                         RON1  0094
BSS 16                                                      RON1  0095
```